Calcul formel

9. Équations différentielles

9.1. Équations différentielles $y' = f(x,y)$

La commande `desolve` permet de résoudre formellement des équations différentielles. Consultez la documentation, accessible par `help(desolve)`.

Travaux pratiques 1.

1. (a) Résoudre l'équation différentielle :
$$y' = y + x + 1.$$
Il s'agit donc de trouver toutes les fonctions $x \mapsto y(x)$ dérivables, telles que $y'(x) = y(x) + x + 1$, pour tout $x \in \mathbb{R}$.
   (b) Résoudre l'équation différentielle en ajoutant une condition initiale :
$$y' = y + x + 1 \quad \text{et} \quad y(0) = 1.$$
2. (a) Résoudre sur $\mathbb{R}_+^*$ l'équation différentielle :
$$x^2 y' = (x-1)y.$$
   (b) Trouver la solution vérifiant $y(1) = 2$.
   (c) Peut-on trouver une solution sur $\mathbb{R}$?
3. Calculer la *fonction logistique*, solution de l'équation différentielle
$$y' = y(1-y) - 1.$$

1. (a) On commence par déclarer la variable x. On définit une fonction $x \mapsto y(x)$ ainsi que sa dérivée $y'(x)$ par `diff(y,x)`. On prendra ici la convention de noter y' par `yy`. L'équation différentielle $y' = y + x + 1$ s'écrit donc `yy == y+x+1`.

Code 1 (*equadiff-intro.sage (1)*).
```
var('x')
y = function('y')(x)
yy = diff(y,x)
desolve(yy == y+x+1, y)
desolve(yy == y+x+1, y, ics=[0,1])
```

On lance la résolution de l'équation différentielle par un appel à la fonction `desolve`, avec pour argument l'équation différentielle à résoudre, ainsi que la fonction inconnue, qui est ici y. Sage renvoie :
`-((x + 1)*e^(-x) - _C + e^(-x))*e^x`
où `_C` désigne une constante arbitraire. Après simplification évidente, les solutions sont les :
$$y(x) = -x - 2 + C\exp(x) \quad \text{où} \quad C \in \mathbb{R}.$$

(b) Il est possible de préciser une condition initiale grâce à l'argument optionnel `ics` (voir la dernière ligne du code ci-dessus), ici on spécifie donc que l'on cherche la solution vérifiant $y(0) = 1$. L'unique solution est alors : $y(x) = -x - 2 + 3\exp(x)$, c'est-à-dire correspondant à $C = 3$.

2. (a) En se restreignant à $\mathbb{R}_+^*$ ($x > 0$), on trouve $y(x) = Cx\exp(\frac{1}{x})$, $C \in \mathbb{R}$.

Code 2 (*equadiff-intro.sage (2)*).
```
equadiff = x^2*yy == (x-1)*y
assume(x>0)
desolve(equadiff, y)
desolve(equadiff, y, ics=[1,2])
```

(b) La solution vérifiant $y(1) = 2$ est $y(x) = 2x\exp(\frac{1}{x} - 1)$, c'est-à-dire $C = 2\exp(-\frac{1}{e})$.

(c) Sur $\mathbb{R}_-^*$ ($x < 0$), les solutions sont aussi de la forme $y(x) = Cx\exp(\frac{1}{x})$.

Code 3 (*equadiff-intro.sage (2)*).
```
forget()
assume(x<0)
desolve(equadiff, y)
```

La seule solution définie sur $\mathbb{R}$ est donc la fonction nulle $y(x) = 0$ (et $C = 0$). En effet, si $C \neq 0$ il n'y a pas de limite finie en 0.

3. Sage résout directement cette équation différentielle.

Code 4 (*equadiff-intro.sage (3)*).
```
equadiff = yy == y*(1-y) - 1
sol_impl = desolve(equadiff, y)
sol = solve(sol_impl,y)
```

en renvoyant :

$$y(x) = -\frac{\sqrt{3}}{2}\tan\left(\frac{\sqrt{3}}{2}C + \frac{\sqrt{3}}{2}x\right) + \frac{1}{2}.$$

9.2. Interprétation géométrique

- Les ***courbes intégrales*** d'une équation différentielle $y' = f(x,y)$ sont les graphes des solutions de cette équation.
- À chaque point d'une courbe intégrale, on associe un vecteur tangent. Si $y(x)$ est une solution de l'équation différentielle alors au point $(x, y(x))$, la tangente est portée par le vecteur $(x', y'(x))$. D'une part $x' = 1$ et d'autre part comme y est solution de l'équation différentielle, on a $y'(x) = f(x,y)$. Ainsi un ***vecteur tangent*** en (x,y) est $\big(1, f(x,y)\big)$.



- Le ***champ de vecteurs*** associé à l'équation différentielle $y' = f(x,y)$ est, en chaque point (x,y) du plan, le vecteur $\big(1, f(x,y)\big)$.
- Voici ce qui signifie géométriquement « résoudre » une équation différentielle. À partir d'un champ de vecteurs, c'est-à-dire la donnée d'un vecteur attaché à chaque point du plan, il s'agit de trouver une courbe intégrale, c'est-à-dire une courbe qui en tout point est tangente aux vecteurs.
- Comme on ne s'occupera pas de la norme des vecteurs, la donnée d'un champ de vecteurs revient ici à associer à chaque point, la pente de la tangente au graphe d'une solution passant par ce point.

Travaux pratiques 2.
Nous allons étudier graphiquement l'équation différentielle

$$y' = -xy.$$

1. Représenter le champ de vecteurs associé à cette équation différentielle.
2. (a) Résoudre l'équation.
   (b) Tracer, sur un même graphe, les courbes intégrales correspondant aux solutions définies par $y(0) = k$, pour quelques valeurs de $k \in \mathbb{R}$.
   (c) Que remarquez-vous ? Quel théorème cela met-il en évidence ?
3. Tracer quelques isoclines de cette équation différentielle.

Les ***isoclines*** de l'équation différentielle $y' = f(x, y)$ sont les courbes sur lesquelles la tangente d'une courbe intégrale a une pente donnée c, c'est-à-dire qu'une isocline est un ensemble

$$\{(x, y) \in \mathbb{R}^2 \mid f(x, y) = c\}.$$

À ne pas confondre avec les solutions ! En chaque point d'une isocline, la solution passant par ce point croise cette isocline avec une pente c.
Voici quelques graphes :
- Le champ de vecteurs (en rouge).
- Les courbes intégrales (en bleu).
- Le champ de vecteurs superposé aux courbes intégrales : les vecteurs sont bien tangents aux courbes intégrales.
- Les isoclines (en vert). Pour une isocline fixée, vérifier que chaque intersection avec les courbes intégrales se fait en des points où la tangente a toujours la même pente.



1. Voici comment tracer le champ de vecteurs associé à $y' = f(x, y)$.

Code 5 (*equadiff-courbe.sage (1)*).
```
def champ_vecteurs(f,xmin,xmax,ymin,ymax,delta):
    G = Graphics()
    for i in srange(xmin, xmax, delta):
        for j in srange(ymin, ymax, delta):
            pt = vector([i,j])
```

```
            v = vector([1,f(u=i,v=j)])
            G = G + arrow(pt,pt+v)
    return G
```

- Les variables de la fonction f sont ici u et v pour ne pas interférer avec la variable x et la fonction y.
- N'hésitez pas à renormaliser les vecteurs v, (par exemple en $v/10$) pour plus de lisibilité.
- En fait la fonction `plot_vector_field`, prédéfinie dans Sage, trace aussi les champs de vecteurs.

2. (a) Les solutions de l'équation sont les

$$y(x) = C\exp\left(-\frac{x^2}{2}\right) \quad \text{où} \quad C \in \mathbb{R}.$$

(b) Le code suivant résout l'équation différentielle et trace la solution avec la condition initiale $y(0) = k$, pour k variant de `kmin` à `kmax` avec un pas de `delta`.

Code 6 (*equadiff-courbe.sage (2)*).

```
def courbes_integrales(equadiff,a,b,kmin,kmax,delta):
    G = Graphics()
    for k in srange(kmin, kmax, delta):
        sol = desolve(equadiff, y, ics=[0,k])
        G = G + plot(sol, (x, a, b))
    return G
```

(c) Les courbes intégrales forment une partition de l'espace : par un point il passe exactement une courbe intégrale. C'est une conséquence du théorème d'existence et d'unicité de Cauchy-Lipschitz.

3. Les isoclines sont définies par l'équation $f(x,y) = c$. Elles se tracent par la commande

`implicit_plot(f-c, (x,xmin,xmax), (y,ymin, ymax))`

Ici les isoclines ont pour équation $-xy = c$, ce sont donc des hyperboles.

9.3. Équations différentielles du second ordre

Travaux pratiques 3.
L'équation différentielle

$$x''(t) + f\,x'(t) + \frac{k}{m}x(t) = 0$$

correspond au mouvement d'une masse m attachée à un ressort de constante $k > 0$ et des frottements $f \geqslant 0$.



Résoudre et tracer les solutions pour différentes valeurs de $f \geqslant 0$ (prendre $k = 1$, $m = 1$), avec les conditions initiales :

$$x(0) = 1 \quad \text{et} \quad x'(0) = 0.$$

Pour une valeur de f, l'équation se résout par :

`desolve(diff(x,t,2) + f*diff(x,t) + k/m*x(t) == 0, x, ics=[0,1,0])`

Voici les solutions pour $f = 0$, $f = 1$ et $f = 2$:





- Cas $f = 0$. Il n'y a pas de frottement. L'équation différentielle s'écrit $x'' + \frac{k}{m}x = 0$. Pour $k = 1$, $m = 1$ alors l'équation est simplement $x'' + x = 0$ dont les solutions sont de la forme
$$x(t) = \lambda \cos t + \mu \sin t \qquad \lambda, \mu \in \mathbb{R}.$$
Avec les conditions initiales $x(0) = 1$ et $x'(0) = 0$, l'unique solution est
$$x(t) = \cos t.$$
Il s'agit donc d'un mouvement périodique.
- Cas $0 < f < 2$. Les frottements sont faibles. Par exemple pour $f = 1$ la solution est
$$\left(\frac{\sqrt{3}}{3}\sin\left(\frac{\sqrt{3}}{2}t\right) + \cos\left(\frac{\sqrt{3}}{2}t\right)\right)e^{-\frac{1}{2}t}.$$
Il s'agit d'un mouvement amorti oscillant autour de la position d'origine $x = 0$.
- Cas $f \geqslant 2$. Les frottements sont forts. Par exemple pour $f = 2$ la solution est
$$x(t) = (t+1)e^{-t}.$$
Il n'y a plus d'oscillations autour de la position d'origine.