Calcul formel

7. Calculs d'intégrales

7.1. Sage comme une super-calculatrice

Travaux pratiques 1.
Calculer à la main les primitives des fonctions suivantes. Vérifier vos résultats à l'ordinateur.

1. $f_1(x) = x^4 + \frac{1}{x^2} + \exp(x) + \cos(x)$
2. $f_2(x) = x\sin(x^2)$
3. $f_3(x) = \frac{\alpha}{\sqrt{1-x^2}} + \frac{\beta}{1+x^2} + \frac{\gamma}{1+x}$
4. $f_4(x) = \frac{x^4}{x^2-1}$
5. $f_5(x) = x^n \ln x$ pour tout $n \geqslant 0$

Pour une fonction f, par exemple `f(x) = x*sin(x^2)` donnée, la commande `integral(f(x),x)` renvoie une primitive de f. Le résultat obtenu ici est `-1/2*cos(x^2)`.
Quelques remarques :

- La machine renvoie une primitive. Pour avoir l'ensemble des primitives, il faut bien entendu ajouter une constante.
- La fonction `log` est la fonction logarithme népérien usuel, habituellement notée ln.
- `integral(1/x,x)` renvoie `log(x)` alors que $\int \frac{dx}{x} = \ln|x|$. Sage omet les valeurs absolues, ce qui ne l'empêche pas de faire des calculs exacts même pour les valeurs négatives (en fait Sage travaille avec le logarithme complexe).
- Souvent, pour intégrer des fractions rationnelles, on commence par écrire la décomposition en éléments simples. C'est possible avec la commande `partial_fraction`. Par exemple, pour `f = x^4/(x^2-1)` la commande `f.partial_fraction(x)` renvoie la décomposition : $f(x) = x^2 + 1 - \frac{1}{2}\frac{1}{x+1} + \frac{1}{2}\frac{1}{x-1}$.

Travaux pratiques 2.
Calculer à la main et à l'ordinateur les intégrales suivantes.

1. $I_1 = \displaystyle\int_1^3 x^2 + \sqrt{x} + \frac{1}{x} + \sqrt[3]{x}\,dx$
2. $I_2 = \displaystyle\int_0^{\frac{\pi}{2}} \sin x(1+\cos x)^4\,dx$
3. $I_3 = \displaystyle\int_0^{\frac{\pi}{6}} \sin^3 x\,dx$
4. $I_4 = \displaystyle\int_0^{\sqrt{3}} \frac{3x^3 - 2x^2 - 5x - 6}{(x+1)^2(x^2+1)}\,dx$

Par exemple `integral(sin(x)^3,x,0,pi/6)` renvoie $\frac{2}{3} - \frac{3\sqrt{3}}{8}$.

7.2. Intégration par parties

Travaux pratiques 3.
Pour n entier positif ou nul, nous allons déterminer une primitive de la fonction

$$f_n(x) = x^n \exp(x).$$

1. Sage sait-il répondre à cette question ?
2. Calculer une primitive F_n de f_n pour les premières valeurs de n.
3. (a) Émettre une conjecture reliant F_n et F_{n-1}.
   (b) Prouver cette conjecture.
   (c) En déduire une fonction récursive qui calcule F_n.
4. (a) Émettre une conjecture pour une formule directe de F_n.
   (b) Prouver cette conjecture.
   (c) En déduire une expression qui calcule F_n.

1. La commande `integral(x^n*exp(x),x)` renvoie le résultat `(-1)^n*gamma(n + 1, -x)`. Nous ne sommes pas tellement avancés ! Sage renvoie une *fonction spéciale* Γ que l'on ne connaît pas. De plus, dériver cette primitive avec Sage (pour la version courante) ne redonne pas la même expression que f_n.
2. Par contre, il n'y a aucun problème pour calculer les primitives F_n pour les premières valeurs de n. On obtient :

$$\begin{aligned}
F_0(x) &= \exp(x)\\
F_1(x) &= (x-1)\exp(x)\\
F_2(x) &= (x^2-2x+2)\exp(x)\\
F_3(x) &= (x^3-3x^2+6x-6)\exp(x)\\
F_4(x) &= (x^4-4x^3+12x^2-24x+24)\exp(x)\\
F_5(x) &= (x^5-5x^4+20x^3-60x^2+120x-120)\exp(x)
\end{aligned}$$

3. (a) Au vu des premiers termes, on conjecture $F_n(x) = x^n \exp(x) - nF_{n-1}(x)$. On vérifie facilement sur les premiers termes que cette conjecture est vraie.
   Sage ne sait pas (encore) vérifier formellement cette identité (avec n une variable formelle). La commande suivante échoue à trouver 0 :
   ```
   integral(x^n*exp(x),x) - x^n*exp(x) + n*integral(x^(n-1)*exp(x),x)
   ```
   (b) Nous prouverons la validité de cette formule en faisant une intégration par parties (on pose par exemple $u = x^n$, $v' = \exp(x)$ et donc $u' = nx^{n-1}$, $v = \exp(x)$) :

$$F_n(x) = \int x^n \exp(x)\,\mathrm{d}x = \left[x^n \exp(x)\right] - \int nx^{n-1}\exp(x)\,\mathrm{d}x = x^n\exp(x) - nF_{n-1}(x).$$

   (c) Voici l'algorithme récursif pour le calcul de F_n utilisant la relation trouvée ci-dessus.

   Code 1 (*integrales-ipp (1)*).
   ```
   def FF(n):
       if n==0:
           return exp(x)
       else:
           return x^n*exp(x) - n*FF(n-1)
   ```

4. Avec un peu de réflexion, on conjecture la formule

$$F_n(x) = \exp x \sum_{k=0}^{n} (-1)^{n-k} \frac{n!}{k!} x^k$$

   où $\frac{n!}{k!} = n(n-1)(n-2)\cdots(k+1)$.
   Pour la preuve, on pose $G_n(x) = \exp(x)\sum_{k=0}^{n}(-1)^{n-k}\frac{n!}{k!}x^k$. On a

$$G_n(x) = \exp(x)\sum_{k=0}^{n-1}(-1)^{n-k}\frac{n!}{k!}x^k + \exp(x)x^n = -nG_{n-1}(x) + \exp(x)x^n.$$

Ainsi G_n vérifie la même relation de récurrence que F_n. De plus $G_0(x) = \exp x = F_0(x)$. Donc pour tout n, $G_n = F_n$. On peut donc ainsi calculer directement notre intégrale par la formule :

```
exp(x)*sum((-1)^(n-k)*x^k*factorial(n)/factorial(k),k,0,n)
```

7.3. Changement de variable

Soit f une fonction définie sur un intervalle I et $\varphi : J \to I$ une bijection de classe $\mathscr{C}^1$. La formule de changement de variable pour les primitives est :

$$\int f(x)\,\mathrm{d}x = \int f\big(\varphi(u)\big)\cdot\varphi'(u)\,\mathrm{d}u.$$

La formule de changement de variable pour les intégrales, pour tout $a, b \in I$, est :

$$\int_a^b f(x)\,\mathrm{d}x = \int_{\varphi^{-1}(a)}^{\varphi^{-1}(b)} f\big(\varphi(u)\big)\cdot\varphi'(u)\,\mathrm{d}u.$$

Travaux pratiques 4.

1. Calcul de la primitive $\displaystyle\int \sqrt{1-x^2}\,\mathrm{d}x$.
   (a) Poser $f(x) = \sqrt{1-x^2}$ et le changement de variable $x = \sin u$, c'est-à-dire poser $\varphi(u) = \sin u$.
   (b) Calculer $f\big(\varphi(u)\big)$ et $\varphi'(u)$.
   (c) Calculer la primitive de $g(u) = f\big(\varphi(u)\big)\cdot\varphi'(u)$.
   (d) En déduire une primitive de $f(x) = \sqrt{1-x^2}$.
2. Calcul de la primitive $\displaystyle\int \frac{\mathrm{d}x}{1+\left(\frac{x+1}{x}\right)^{1/3}}$.
   (a) Poser le changement de variable défini par l'équation $u^3 = \frac{x+1}{x}$.
   (b) En déduire le changement de variable $\varphi(u)$ et $\varphi^{-1}(x)$.
3. Écrire une fonction `integrale_chgtvar(f,eqn)` qui calcule une primitive de f par le changement de variable défini par l'équation `eqn` reliant u et x.
   En déduire $\displaystyle\int (\cos x + 1)^n\cdot\sin x\,\mathrm{d}x$ (pour $n > 0$) en posant $u = \cos x + 1$.
4. Même travail avec `integrale_chgtvar_bornes(f,a,b,eqn)` qui calcule l'intégrale de f entre a et b par changement de variable.
   En déduire $\displaystyle\int_0^{\frac{\pi}{4}} \frac{\mathrm{d}x}{2+\sin^2 x}$ en posant $u = \tan x$.

1. (a) La fonction est $f(x) = \sqrt{1-x^2}$ et on pose $\varphi(u) = \sin u$, c'est-à-dire que l'on espère que le changement de variable $x = \sin u$ va simplifier le calcul de l'intégrale :

```
f = sqrt(1-x^2)     phi = sin(u)
```

   (b) On obtient $f\big(\varphi(u)\big)$ en substituant la variable x par $\sin u$: cela donne $f\big(\varphi(u)\big) = \sqrt{1-\sin^2 u} = \sqrt{\cos^2 u} = |\cos u|$. Ce qui s'obtient par la commande `frondphi = f(x=phi)` (puis en simplifiant). Noter que Sage « oublie » les valeurs absolues, car il calcule en fait la racine carrée complexe et non réelle. Ce sera heureusement sans conséquence pour la suite. En effet pour que $f(x)$ soit définie, il faut $x \in [-1, 1]$, comme $x = \sin u$ alors $u \in [-\frac{\pi}{2}, \frac{\pi}{2}]$ et donc $\cos u \geqslant 0$. Enfin $\varphi'(u) = \cos u$ s'obtient par `dphi = diff(phi,u)`.

   (c) On pose $g(u) = f\big(\varphi(u)\big)\cdot\varphi'(u) = \cos^2 u = \frac{1+\cos(2u)}{2}$. Donc $\int g(u)\,\mathrm{d}u = \frac{u}{2} + \frac{\sin(2u)}{4}$.

   (d) Pour obtenir une primitive de f, on utilise la formule de changement de variable :

$$\int \sqrt{1-x^2}\,\mathrm{d}x = \int f(x)\,\mathrm{d}x = \int g(u)\,\mathrm{d}u = \frac{u}{2} + \frac{\sin(2u)}{4}.$$

   Mais il ne faut pas oublier de revenir à la variable x. Comme $x = \sin u$ alors $u = \arcsin x$, donc

$$\int \sqrt{1-x^2}\,\mathrm{d}x = \frac{\arcsin x}{2} + \frac{\sin(2\arcsin x)}{4}.$$

Les commandes sont donc `G = integral(g,u)` puis `F = G(u = arcsin(x))` pour obtenir la primitive recherchée. Après simplification de $\sin(2\arcsin x)$, on obtient :

$$\int \sqrt{1-x^2}\,\mathrm{d}x = \frac{\arcsin x}{2} + \frac{x}{2}\sqrt{1-x^2}.$$

2. Pour calculer la primitive $\int \frac{\mathrm{d}x}{1+\left(\frac{x+1}{x}\right)^{1/3}}$, l'énoncé nous recommande le changement de variable $u^3 = \frac{x+1}{x}$. La seule difficulté supplémentaire est qu'il faut exprimer x en fonction de u et aussi u en fonction de x. Pour cela, on définit l'équation `u^3 == (x+1)/x` reliant u et x : `eqn = u^3 == (x+1)/x`. On peut maintenant résoudre l'équation afin d'obtenir $\varphi(u)$ (x en fonction de u) : `phi = solve(eqn,x)[0].rhs()` et $\varphi^{-1}(x)$ (u en fonction de x) : `phi_inv = solve(eqn,u)[0].rhs()`. Le reste se déroule comme précédemment.

3. On automatise la méthode précédente :

Code 2 (*integrales-chgtvar (1)*).
```
def integrale_chgtvar(f,eqn):
    phi = solve(eqn,x)[0].rhs()     # x en fonction de u : fonction phi(u)=x
    phi_inv = solve(eqn,u)[0].rhs() # u en fonction de x : inverse de phi : phi_inv(x)=u
    frondphi = f(x=phi)             # la fonction f(phi(u))
    dphi = diff(phi,u)              # sa dérivée
    g = frondphi*dphi               # la nouvelle fonction g(u)
    g = g.simplify_full()
    G = integral(g,u)               # g doit être plus facile à intégrer
    F = G(u = phi_inv)              # on revient à la variable x
    F = F.simplify_full()
    return F
```

Ce qui s'utilise ainsi :

Code 3 (*integrales-chgtvar (2)*).
```
f = (cos(x)+1)^n*sin(x)
eqn = u == cos(x)+1
integrale_chgtvar(f,eqn)
```

et montre que

$$\int (\cos x + 1)^n \cdot \sin x \,\mathrm{d}x = -\frac{1}{n+1}(\cos x + 1)^{n+1}.$$

4. Pour la formule de changement de variable des intégrales, on adapte la procédure précédente en calculant l'intégrale $\int_{\varphi^{-1}(a)}^{\varphi^{-1}(b)} g(u)\,\mathrm{d}u$.
Pour cela, on calcule $\varphi^{-1}(a)$ par `a_inv = phi_inv(x=a)` et $\varphi^{-1}(b)$ par `b_inv = phi_inv(x=b)`.
Pour calculer $\int_0^{\frac{\pi}{4}} \frac{\mathrm{d}x}{2+\sin^2 x}$, on pose $u = \tan x$. Donc $x = \varphi(u) = \arctan u$ et $u = \varphi^{-1}(x) = \tan x$. Nous avons $\varphi'(u) = \frac{1}{1+u^2}$. D'autre part, comme $a = 0$ on a $\varphi^{-1}(a) = \tan 0 = 0$ et comme $b = \frac{\pi}{4}$ on a $\varphi^{-1}(b) = \tan\frac{\pi}{4} = 1$. Ainsi la formule de changement de variable :

$$\int_a^b f(x)\,\mathrm{d}x = \int_{\varphi^{-1}(a)}^{\varphi^{-1}(b)} f\big(\varphi(u)\big)\cdot\varphi'(u)\,\mathrm{d}u$$

devient

$$I = \int_0^{\frac{\pi}{4}} \frac{1}{2+\sin^2 x}\,\mathrm{d}x = \int_0^1 \frac{1}{2+\sin^2(\arctan u)}\frac{1}{1+u^2}\,\mathrm{d}u.$$

Mais $\sin^2(\arctan u) = \frac{u^2}{1+u^2}$ donc

$$I = \int_0^1 \frac{1}{2+\frac{u^2}{1+u^2}}\frac{1}{1+u^2}\,\mathrm{d}u = \int_0^1 \frac{\mathrm{d}u}{2+3u^2}\,\mathrm{d}u = \left[\frac{1}{\sqrt{6}}\arctan\left(\frac{\sqrt{6}u}{2}\right)\right]_0^1 = \frac{1}{\sqrt{6}}\arctan\left(\frac{\sqrt{6}}{2}\right).$$

Ce que l'ordinateur confirme !